Tutorial 2 MT264

Tutor : Rifat Hamoudi
Staff No. : 00567451
Mobile No. : 0781-2796265

I have put this tutorial on the web. This tutorial can be viewed and downloaded from http://www.rifathamoudi.co.uk then selecting MT264 Tutorials then Tutorial 2.
1) Explain with examples what the followings are :

a) Container

b) Controls

2) 

(a) You have been asked to design a form that can be used for online ordering of shoes from a catalogue. The form contains a picture to identify a style of shoe. Choose controls to enable the customer to select the colour and size of the shoes they require in that style, as well as the number of pairs required.  The price of a pair of shoes is to be displayed, as well as the total cost of the order. Finally, there should be a means for the customer to submit the order (add to a shopping basket, perhaps) when it is complete. For each of these pieces of data, state what data types you feel are suitable, and explain what controls you think would be appropriate to use on the form.

(b) Discuss the layout of the ShoeOrderForm part (a), paying attention to the design principles from section 4 of unit 1.

(c) Write out property tables for the ShoeOrderForm in question 2.

(d) Write out an event table for this application. Include comments to describe what each event handler does.

3) Design and implement a Visual Basic program that adds two numbers. An example output is as shown below :

4) Design and implement a Visual Basic program that counts the number of letters in a string. An example output is as shown below :

5) Design and implement a Visual Basic program that adds two numbers entered by the user. An example output is as shown below :

6) Design and implement functional code for a project entitled “Traffic Survey” with the following specification:

An application is required for collecting information about traffic from a particular location. More precisely, we wish to record the number of cars, bicycles and lorries passing (in either direction) a particular point at the roadside. The idea is that the user will record each car, bicycle and lorry as they pass. The application will maintain and display the total numbers of each type of vehicle.

7) Design a GUI for a simple database that can be used to keep track of items by letting the user enter the item ID and the details. Make the design simple but effective.

Design and code a fully functional simple database software according to the following specification.

Project Specification : Simple Database

The simple database should have :

a) way to enter new item identifiers and item descriptions in the software

b) way to display the entered items

c) way to save the items entered

d) way to retrieve the items entered without having to re-enter them

e) add instructions/help to the software

f) add the About part for this software

Show the software design, the GUI design and the implementation

Assume that this software is designed for a small sized startup company. Test the software by entering the name of the employee as item identifier and the address as the description. This way we should have a simple database model to keep track of all the employees’ details in the company.

Answer to Question 1

The Form “FixedAdder” is a Container
There are 7 Controls contained within it.

First Number, Second Number and Total are called Label controls

First Number Label control has Text property initialised to 34

Second Number Label control has Text property initialised to 26

Total Label control has Text property initialised to “Result”

Button control is initialised to “Add”

Answer to Question 2

(a)

	Control
	Control Type
	Underlying Data type

	colour
	ComboBox
	String

	size
	ComboBox 
	Integer*

	number of pairs
	NumericUpDown
	Integer

	price
	TextBox 
	Double

	total cost
	TextBox 
	Double

	submit order
	Button
	-

(b)

(c)

	
	Name
	Property
	Initial Value

	Container
	
	
	

	Form
	MainForm
	Text
	Shoe Order

	Controls
	
	
	

	Label
	colourLabel
	Text
	Colour

	ComboBox
	colourComboBox
	Items
	Black, Brown, Black Patent

	Label
	sizeLabel
	Text
	Size

	ComboBox
	sizeComboBox
	Items
	7, 8, 9, 10, 11, 12, 13, 14

	NumericUpDown
	quantityNumericUpDown
	Value
	0

	
	
	Minimum
	0

	
	
	Maximum
	10

	Label
	priceLabel
	Text
	Price £s

	TextBox
	priceTextBox
	Text
	

	Label
	totalLabel
	Text
	Total Cost £s

	TextBox
	totalTextBox
	Text
	

	Button
	submitButton
	Text
	Submit Order


(d) Event Table: Shoe Order
-------------------------------------------------------------------------------------------------------

colorComboBox  selected index changed

OnSelectedIndexChanged


'Display colour by updating the view

-------------------------------------------------------------------------------------------------------

sizeComboBox selectedIndex changed

OnSelectedIndexChanged


'Display size and price by updating the view

--------------------------------------------------------------------------------------------------------

quantityNumericUpDown value changed

OnValueChanged


'Display quantity and total cost by updating the view

--------------------------------------------------------------------------------------------------------submitButton clicked

OnClick


'submit order

'update view to show all fields reset

-------------------------------------------------------------------------------------------------------

Answer to Question 3

Public Class AddForm

    Private Sub AddButton_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles AddButton.Click

Dim FirstNum As Integer

      Dim SecondNum As Integer

        FirstNum = Num1.Text

        SecondNum = Num2.Text

        ' A statement to add user input values and assign the total.

        Sum.Text = FirstNum + SecondNum
    End Sub

End Class

Implementation of the answer to question 3

1) File then New Project

Choose “Windows Forms Application”

2) 

Click on Toolbar and fix it

3) In this example choose 6 Label controls and 1 Button control and drag that onto the Form1 as follows :

4) For the first 3 controls put the names by changing them from the Text property in the Label object. In Label4 change Text to a number for example 34. In the Button1 object change the Text to “Add” and the Name to “AddButton”.

5) Double click on the Add button and you get the event handler and add the following statement :

Dim FirstNum As Integer

      Dim SecondNum As Integer

        FirstNum = Num1.Text

        SecondNum = Num2.Text

        ' A statement to add user input values and assign the total.

        Sum.Text = FirstNum + SecondNum
So the final code should look like this :

Public Class AddForm

    Private Sub AddButton_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles AddButton.Click

Dim FirstNum As Integer

      Dim SecondNum As Integer

        FirstNum = Num1.Text

        SecondNum = Num2.Text

        Sum.Text = FirstNum + SecondNum
    End Sub

End Class

6) Press F5 or the green button to load the software. Click on the “Add” button and you should see the addition of the 2 numbers
Answer to Question 4

Public Class Form1

    Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button1.Click

        ' Display the character count of the textbox.

        MsgBox("You typed: " & Str(Len(TextBox1.Text)) & " characters")

    End Sub

End Class

The above code is the efficient type hence it is shorter but less readable. A less efficient (but more readable) code is as follows :

        Dim Lengthofstring As Integer

        Dim StringLength As String

        Lengthofstring = TextBox1.Text.Length

        StringLength = Str(Lengthofstring)

        MsgBox("You typed: " & StringLength & " characters")

But the most efficient code is as follows:
Public Class Form1

    Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Button1.Click

        ' Display the character count of the textbox.

        MsgBox("You typed: " & TextBox1.Text.Length.ToString & " characters")

    End Sub

End Class
Answer to Question 5

Public Class Form1

    Private Sub AddBtn_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles AddBtn.Click

Dim FirstNum As Integer

      Dim SecondNum As Integer

        FirstNum = Num1.Text

        SecondNum = Num2.Text

        ' A statement to add user input values and assign the total.

        Sum.Text = FirstNum + SecondNum    

End Sub

    Private Sub ClearBtn_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles ClearBtn.Click

        ' Statements to resume the initial state.

        Sum.Text = "Result" : Num1.Text = "" : Num2.Text = ""

    End Sub

End Class

Answer to Question 6

Software design

Public Class TrafficS
    ' Declare the fields here.
    Private fCars As Integer
    Private fBicycles As Integer
    Private fLorries As Integer
    Public Sub New()

   ' An instance of Traffic is created with all vehicle counts set to zero.
        fCars = 0

        fBicycles = 0

        fLorries = 0

    End Sub
    ' Setters
    Public Sub addCar()

        fCars = fCars + 1

    End Sub
    Public Sub addBicycle()

        fBicycles = fBicycles + 1

    End Sub
    Public Sub addLorry()

        fLorries = fLorries + 1

    End Sub
    'Getters
    Public ReadOnly Property getCar() As Integer
        Get
            Return fCars

        End Get
    End Property
    Public ReadOnly Property getBicycle() As Integer
        Get
            Return fBicycles

        End Get
    End Property
    Public ReadOnly Property getLorry() As Integer
        Get
            Return fLorries

        End Get
    End Property
End Class

Public Class MainForm
    Private fTrafficSurvey As TrafficS
    Public Sub New()

        fTrafficSurvey = New TrafficS
        ' This call is required by the designer.
        InitializeComponent()

        ' Add any initialization after the InitializeComponent() call.
        updateView()

    End Sub
    Private Sub CarsCount_Click(sender As System.Object, e As System.EventArgs) Handles CarsCount.Click

        fTrafficSurvey.addCar()

        Car.Text = fTrafficSurvey.getCar()

    End Sub
    Private Sub BicyclesCount_Click(sender As System.Object, e As System.EventArgs) Handles BicyclesCount.Click

        fTrafficSurvey.addBicycle()

        Bicycle.Text = fTrafficSurvey.getBicycle()

    End Sub
    Private Sub LorriesCount_Click(sender As System.Object, e As System.EventArgs) Handles LorriesCount.Click

        fTrafficSurvey.addLorry()

        Lorry.Text = fTrafficSurvey.getLorry()

    End Sub
    Private Sub updateView()

        Car.Text = fTrafficSurvey.getCar()

        Bicycle.Text = fTrafficSurvey.getBicycle()

        Lorry.Text = fTrafficSurvey.getLorry()

    End Sub
    Private Sub AddVehicles_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles AddVehicles.Click

        Total.Text = fTrafficSurvey.getCar() + fTrafficSurvey.getBicycle() + fTrafficSurvey.getLorry()

    End Sub
    Private Sub ExitToolStripMenuItem_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles ExitToolStripMenuItem.Click

        'The application is closed.
        Me.Close()

    End Sub
End Class

Answer to Question 7

For the main form we can design something like the following :

Here we have :

1) MenuStrip

2) ComboBox

3) TextBox

4) Button

It is better to split the design so that when the user click on the Add Item button they will get another form as follows :

[image: image21.png]K& SimpleDB - Microsoft Yisual Basic 2008 Express Edi

Fle Edt Vew Projct Buld Debug Data Fomat Took Window el

GEE - D@ s BB =2

9-0c QB ouom SE(

[E@R B

B
SimpleDatabase.vb_| NenEriry.vb | SimpleDB.vh | SimpleDB.vb [Design] | AboutSimpleDE.vh [Design]**NewEntry.vb [Design] s x

5] Directoryeriry

2 Drectorysearcher

% Domainusboun

O Enorproder

8] Evention

53l Flesystemwatcher

= FlowlayoutPanel

] Folderronserdisog

£ FortDidog

) GrousBox

HelpProvider

> HrolBar

5 ngetst

A Label

A Linkiabel

=3 ustpox

23 Listview

(o] MaskedTestBox

= enustp

3 MessageQueue

7% Monthcalendar
Not#ytcon

143 Nmericupboun

27 OperFieDibog

£ PagesetupDion

] Parel

2] PerformanceCounter

4l FcureBoe

Enterltem D

Etertem Desciption

Eror st ~ %
(@ 0Errrs | [L1,2 Warnings | [ (D) 0 Messages
Descrption e e Column | Project
51 Variable ‘sReader s sed before R has been assined 2 value, &l reerence sxception could esul t runtme, SmpleDatabase b 95 16 SimpleDE
42 Variable ‘airter is used before i has been assined a value A null rference exception could esul at rurime. SmpleDatabase b 124 16 SimpleDB

Wy project
AboutSinpleDB.vb

NewEntry b

) SinpleDatabase. b

SimpleDB.vb

5] smpleD_instructions.vb

goktion xplrer [Tosts Sources

[Poperties <X

Newntry System. Windows. Forms.F ~

=l > 1=

Maxinunsize 0,0 A

MinimizeBox  False
Miimursize 0,0
Opacity 100%
Paddng 0,0,0,0
RightToleft  ho
RightToLeftLayo False
Showlcon  True
ShawlnTaskbar False

ES 375,340
SeeGrpstyle  Auto
StartPostion  CenterParent
Tag

Text Newntry

Text
The text associated with the contral

Ready




Here we have 2 textboxes, the bottom one is multilane

Software design



GUI Design

Main Form : SimpleDB.vb

Property Table : Display the entered items

	
	Name
	Property
	Initial Value



	Container
	
	
	

	Form
	SimpleDB
	Text
	Display the entered items

	Controls
	
	
	

	ComboBox
	itemComboBox
	
	Scroll through the items entered

	TextBox
	definitionTextBox
	Text
	Show the properties of the items entered

	MenuStrip
	MenuStrip1
	File
	This is to show the File properties that includes : New, Open, Save, Save As and Exit

	MenuStrip
	MenuStrip1
	Help
	This is to show the Help properties that includes : Instructions and About information

	Button
	AddItemBut
	Text
	Loading the New Entry Form


Entry Form : NewEntry.vb

Property Table : Enter new items

	
	Name
	Property
	Initial Value



	Container
	
	
	

	Form
	NewEntry
	Text
	Enter new items

	Controls
	
	
	

	TextBox
	itemTextBox
	Text
	Enter the Item’s identifier

	TextBox
	descriptionTextBox
	Text
	Enter the Item’s description

	Button
	OK_Button
	Text
	OK to enter the item’s details

	Button
	Cancel_Button
	Text
	Cancel


Instructions Form : SimpleDB_Instructions

Property Table : Enter new items

	
	Name
	Property
	Initial Value



	Container
	
	
	

	Form
	SimpleDB_Instructions
	Text
	Display the instructions

	Controls
	
	
	

	TextBox
	instructTextBox
	Text
	Display the instructions

	Button
	OK_Button
	Text
	OK to close the form


About DialogBox : AboutSimpleDB

[image: image27.png]K& SimpleD
Bl Edt Vew Froject

] DvectoryEry
2 Drectorysearcher
1 DomainpDonn
D Eorprovider

8] Evertiog

53l Flesystemwatcher
= FlowlayoutPanel
3] FlderbrowserDisog
£ FortDisog

"] Grouptox
HelpProvider

<> scrolger

P magelist

A Label

A Linkiabel

=3 ustpox

23 Listview

(o] MaskedTestBox
Venustn

3 MessageQueue
7] MorthClendar
Notfylcon

122 umericupDonn
27 OperFieDibog

] PagesetupDiabog
[] panel

2] PerformanceCounter
4] Pitweox

icrosoft Visual Bas

Buld Debug Data

HEE-Hd s RR|=2

2008 Express Edi

Fomat  Took Window Help

9.0 @-B| > um s

EEEPI-=1

‘AboutSimpleDB.vb [Design] | ewErtry.vb [Desin] | SimpleD_Inst,..nsvh [Design] | SipleDatabase:vb | NewEntry b | SimpleDevb |

Company Name
[Desciption

[t niime, the labels test wil be replaced with
the appiication's assembly infamatian,

Customize the appicalirts assembly information
i the Application pane of Project Designet

Err Lot
(@ 0Errrs | [L1,2 Warnings | [ (D) 0 Messages

Descrption e e Column | Project
51 Variable ‘sReader s sed before R has been assined 2 value, &l reerence sxception could esul t runtme, SmpleDatabase b 83 16 SimpleDE
42 Variable ‘airter is used before i has been assined a value A null rference exception could esul at rurime. SmpleDatabaseh 112 16 SimpleDB

) SimpleDatabase. b
] simpleDB.vb
5] sileDe,_Instructions.vh

goktion xplrer [Tosts Sources

AboutSimpleDB System. Windows.F +
EN[E
Madimumsize 0,0 -
Minimizeox  False
Vinimumsize 0,0

opacty  100%

Paddng  9,9,9,9
RightToleft o
RightToLeftlay Fabe

Shawlcon  True
ShawlnTaskbar False
ES 420,301

SeeGripstyle  Auto
StartPostion  CenterParent
Tag

Text AboutBox

Text
The text associated with the contral

Ready




Property Table : About the software information

	
	Name
	Property
	Initial Value



	Container
	
	
	

	DialogBox
	AboutSimpleDB
	Text
	Show the software’s information

	Controls
	
	
	

	LogoPictureBox
	LogoPictureBox
	Image
	Display the logo picture

	Button
	OK_Button
	Text
	OK to close the form

	TextBox
	LabelProductName
	Text
	Display the software name

	TextBox
	LabelVersion
	Text
	Display the label version

	TextBox
	LabelCopyright
	Text
	Display the copyright

	TextBox
	LabelCompanyName
	Text
	Display the company name


SimpleDatabase software testing

1) Entering new data

Load the software and click on “Add Item” then type the item’s details

2) Displaying new data

Load the software and click on the combo box to display the item’s details

[image: image29.png][ smotenn =Tk}

Ble Hep

Selecttem ID

Showlem Detais

[address 23, Dpen Universy, London, N11
Project: Programmer
Salay  £23.000

Addltem





3) Saving new data

Click on File then Save and give a filename to save the data in. In this case the file is called db.txt

4) Retrieving data

Click on File then Open and give a filename to retrieve the data in. In this case the file is called db.txt

5) Show instructions on how to use the software

Click on Help then Instructions

6) Show About dialog for the SimpleDB software

Click on Help then Instructions

Software implementation

SimpleDatabase class definition and implementation

Imports System.IO

Public Class SimpleDatabase

    Private fItem As Dictionary(Of String, String)

    Private fModified As Boolean

    Private fFileName As String

    Public Sub New()

        'Preconditions: none

        'Postconditions: a SimpleDatabase object has been created with the

        'glossary initially empty, FileName set to "" and Modified set to False.

        FileName = ""

        fItem = New Dictionary(Of String, String)

        fModified = False

    End Sub

    Public Property FileName() As String

        Get

            Return fFileName

        End Get

        Set(ByVal value As String)

            fFileName = value

        End Set

    End Property

    Public ReadOnly Property Modified() As Boolean

        Get

            Return fModified

        End Get

    End Property

    Public ReadOnly Property WordList() As Dictionary(Of String, String).KeyCollection

        'Preconditions: none

        'Postconditions: The database item entries are returned as a collection of keys.

        Get

            Return fItem.Keys

        End Get

    End Property

    Public ReadOnly Property Description(ByVal word As String) As String

        'Preconditions: item is an entry in the database

        'Postconditions: The description associated with the item entry 

        'for the item is returned.

        Get

            Return fItem.Item(word)

        End Get

    End Property

    Public Sub add(ByVal word As String, ByVal description As String)

        'Preconditions: item and description are not empty strings and item is not

        'already an entry in the database.

        'Postconditions: A new entry for item with associated description is added.

        'Modified is set to True.

        fItem.Add(word, description)

        fModified = True

    End Sub

    Public Sub load()

        'Preconditions: none

        'Postconditions: If FileName is a valid path for a text file and this file is of

        'appropriate format, then the glossary is loaded from this file. Otherwise the

        'glossary is cleared and an exception is thrown. It is the responsibility of

        'client code to handle the exception. Modified is set to False in either case.

        Dim aReader As StreamReader

        fItem.Clear()

        Try

            aReader = New StreamReader(FileName)

            While Not (aReader.EndOfStream)

                fItem.Add(aReader.ReadLine(), aReader.ReadLine())

            End While

        Catch ex As ArgumentException

            fItem.Clear()

            Throw New ArgumentException("There was a problem opening the file " + FileName)

        Catch ex As IOException

            fItem.Clear()

            Throw New IOException("There was an input-output error for " + FileName)

        Catch ex As UnauthorizedAccessException

            fItem.Clear()

            Throw New UnauthorizedAccessException("There was an unauthorised access error for " + FileName)

        Finally

            fModified = False

            If aReader IsNot Nothing Then

                aReader.Close()

            End If

        End Try

    End Sub

    Public Sub save()

        'Preconditions: none

        'Postconditions: If FileName is a valid path for a text file, then the items of the

        'glossary are stored in the file and Modified is set to False. Otherwise an exception

        'is thrown. It is the responsibility of client code to handle the exception.

        Dim aWriter As StreamWriter

        Try

            aWriter = New StreamWriter(FileName)

            For Each key As String In WordList

                aWriter.WriteLine(key)

                aWriter.WriteLine(Description(key))

            Next

            fModified = False

        Catch ex As ArgumentException

            fItem.Clear()

            Throw New ArgumentException("There was a problem opening the file " + FileName)

        Catch ex As IOException

            fItem.Clear()

            Throw New IOException("There was an input-output error for " + FileName)

        Catch ex As UnauthorizedAccessException

            fItem.Clear()

            Throw New UnauthorizedAccessException("There was an unauthorised access error for " + FileName)

        Finally

            If aWriter IsNot Nothing Then

                aWriter.Close()

            End If

        End Try

    End Sub

End Class

Main Form : SimpleDB.vb

Imports System.IO

Public Class SimpleDB

    Private fSimpleDatabase As SimpleDatabase

    Public Sub New()

        fSimpleDatabase = New SimpleDatabase

        ' This call is required by the Windows Form Designer.

        InitializeComponent()

        ' Add any initialization after the InitializeComponent() call.

        updateView()

    End Sub

    Public Sub updateView()

        itemComboBox.Items.Clear()

        definitionTextBox.Clear()

        For Each key As String In fSimpleDatabase.WordList

            itemComboBox.Items.Add(key)

        Next

    End Sub

    Private Sub AboutToolStripMenuItem_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles AboutToolStripMenuItem.Click

        AboutSimpleDB.ShowDialog()

    End Sub

    Private Sub itemComboBox_SelectedIndexChanged(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles itemComboBox.SelectedIndexChanged

        'The description of the selected word is displayed.

        definitionTextBox.Text = _

      fSimpleDatabase.Description(itemComboBox.SelectedItem.ToString())

    End Sub

    Private Sub ContentsToolStripMenuItem_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles ContentsToolStripMenuItem.Click

        ' This event handler displays a dialog box with the instructions

        ' for playing the game.

        Dim anInstructDialog As SimpleDB_Instructions

        anInstructDialog = New SimpleDB_Instructions

        anInstructDialog.ShowDialog()

        anInstructDialog.Dispose()

    End Sub

    Private Sub saveMenuItem_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles saveMenuItem.Click

        'If the file name is empty, the event handler for saveAsMenuItem is used.

        'Otherwise the glossary is saved and the view is updated. If an exception is 

        'caught, the file name is set to "" and an error message is displayed.

        'Add your code here

        If fSimpleDatabase.FileName = "" Then

            saveAsMenuItem_Click(sender, e)

        Else

            Try

                fSimpleDatabase.save()

                updateView()

            Catch ex As ArgumentException

                fSimpleDatabase.FileName = ""

                MessageBox.Show(ex.Message)

            Catch ex As IOException

                fSimpleDatabase.FileName = ""

                MessageBox.Show(ex.Message)

            Catch ex As UnauthorizedAccessException

                fSimpleDatabase.FileName = ""

                MessageBox.Show(ex.Message)

            Catch ex As Exception

                fSimpleDatabase.FileName = ""

                MessageBox.Show(ex.Message)

            End Try

        End If

    End Sub

    Private Sub saveAsMenuItem_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles saveAsMenuItem.Click

        If (aSaveFileDialog.ShowDialog() = DialogResult.OK) Then

            fSimpleDatabase.FileName = aSaveFileDialog.FileName

            saveMenuItem_Click(sender, e) 'Reuse code to save the file

        End If

    End Sub

    Private Sub ExitToolStripMenuItem_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles ExitToolStripMenuItem.Click

        Close()

    End Sub

    Private Sub openMenuItem_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles openMenuItem.Click

        If fSimpleDatabase.Modified Then

            If MessageBox.Show("Database has been modified, save?", "SimpleDatabase", _

                  MessageBoxButtons.YesNo) = DialogResult.Yes Then

                saveMenuItem_Click(sender, e)  'Reuse code to save the file

            End If

        End If
        If (anOpenFileDialog.ShowDialog() = DialogResult.OK) Then

            Try

                fSimpleDatabase.FileName = anOpenFileDialog.FileName

                fSimpleDatabase.load()

            Catch ex As ArgumentException

                fSimpleDatabase.FileName = ""

                MessageBox.Show(ex.Message)

            Catch ex As IOException

                fSimpleDatabase.FileName = ""

                MessageBox.Show(ex.Message)

            Catch ex As UnauthorizedAccessException

                fSimpleDatabase.FileName = ""

                MessageBox.Show(ex.Message)

            Catch ex As Exception

                fSimpleDatabase.FileName = ""

                MessageBox.Show(ex.Message)

            End Try

        End If

        updateView()

    End Sub

    Private Sub AddItemBut_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles AddItemBut.Click

        'A dialogue box for a new entry is displayed. If an appropriate entry is

        'made and OK is clicked, the entry is added to the glossary. If the entry

        'already exists, a warning is displayed instead. The GUI is updated.

        Dim aNewEntryDialog As NewEntry

        Dim word As String

        Dim description As String

        aNewEntryDialog = New NewEntry

        If (aNewEntryDialog.ShowDialog() = DialogResult.OK) Then

            word = aNewEntryDialog.itemTextBox.Text

            description = aNewEntryDialog.descriptionTextBox.Text

            If fSimpleDatabase.WordList.Contains(word) Then

                MessageBox.Show("Cannot add new entry - item already contained in database")

            Else

                fSimpleDatabase.add(word, description)

            End If

        End If

        aNewEntryDialog.Dispose()

        updateView()

    End Sub

End Class

Entry Form : NewEntry.vb

Imports System.Windows.Forms

Public Class NewEntry

    Private Sub OK_Button_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles OK_Button.Click

        Dim aItem As String

        Dim description As String

        aItem = Me.itemTextBox.Text

        description = Me.descriptionTextBox.Text

        If aItem = "" Or description = "" Then

            Me.DialogResult = Windows.Forms.DialogResult.None

        Else

            Me.DialogResult = System.Windows.Forms.DialogResult.OK

            Me.Close()

        End If

    End Sub

    Private Sub Cancel_Button_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles Cancel_Button.Click

        Me.DialogResult = System.Windows.Forms.DialogResult.Cancel

        Me.Close()

    End Sub

End Class

Instructions Form : SimpleDB_Instructions

Public Class SimpleDB_Instructions

    Private Sub SimpleDB_Instructions_Load(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MyBase.Load

        ' The file to be loaded is plain text, so we have to specify the type.

        instructTextBox.LoadFile("Instructions.txt", RichTextBoxStreamType.PlainText)

    End Sub

    Private Sub OK_Button_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles OK_Button.Click

        Me.DialogResult = System.Windows.Forms.DialogResult.OK

        Me.Close()

    End Sub

End Class

About DialogBox : AboutSimpleDB

Public NotInheritable Class AboutSimpleDB

    Private Sub AboutBox_Load(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MyBase.Load

        ' Set the title of the form.

        Dim ApplicationTitle As String

        If My.Application.Info.Title <> "" Then

            ApplicationTitle = My.Application.Info.Title

        Else

            ApplicationTitle = System.IO.Path.GetFileNameWithoutExtension(My.Application.Info.AssemblyName)

        End If

        Me.Text = String.Format("About {0}", ApplicationTitle)

        ' Initialize all of the text displayed on the About Box.

        ' TODO: Customize the application's assembly information in the "Application" pane of the project 

        '    properties dialog (under the "Project" menu).

        Me.LabelProductName.Text = "Product Name : SimpleDB"

        Me.LabelVersion.Text = String.Format("Version {0}", My.Application.Info.Version.ToString)

        Me.LabelCopyright.Text = "Copyright : Rifat Hamoudi 2009"

        Me.LabelCompanyName.Text = "Company Name : Open University"

        Me.TextBoxDescription.Text = "Description : Basic database to add and retrieve items. This is written using OOP and OOD methodologies."

    End Sub

    Private Sub OKButton_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles OKButton.Click

        Me.Close()

    End Sub

End Class

Traffic Survey Class 

